Chapter 5
INSET Statement

Chapter Table of Contents

Chapter 5
INSET Statement

Overview

Graphical displays such as histograms and probability plots are commonly used for process capability analysis. You can use the INSET statement to enhance these plots by adding a box or table (referred to as an *inset*) of summary statistics directly to the graph. An inset typically displays statistics calculated by the CAPABILITY procedure but can also display values provided in a SAS data set. A typical application of the INSET statement is to augment a histogram with the sample size, mean, standard deviation, and process capability index C_{pk}.

Note that the INSET statement by itself does not produce a display and must be used with the CDFPLOT, COMPHISTOGRAM, HISTOGRAM, PPPLOT, PROBPLOT, or QQPLOT statement. *

You can use options in the INSET statement to

- specify the position of the inset
- specify a header for the inset table
- specify graphical enhancements, such as background colors, text colors, text height, text font, and drop shadows

*In Release 6.12 and in previous releases of SAS/QC software, the keyword GRAPHICS was required in the PROC CAPABILITY statement since the INSET statement only enhances output created on high resolution graphics devices.

Getting Started

This section introduces the INSET statement with examples that illustrate commonly used options. Complete syntax for the INSET statement is presented in the "Syntax" section on page 196, and advanced examples are given in the "Examples" section on page 211.

Displaying Summary Statistics on a Histogram

See CAPINS1 in the SAS/QC Sample Library

In a plant producing copper wire, an important quality characteristic is the torsion strength, measured as the twisting force in pounds per inch necessary to break the wire. The following statements create the SAS data set WIRE, which contains the torsion strengths (STRENGTH) for 50 different wire samples:

```
data wire;
   label strength='Torsion Strength in lb/in';
   input strength @@;
   datalines;
25 25 36 31 26 36 29 37 37 20
34 27 21 35 30 41 33 21 26 26
19 25 14 32 30 29 31 26 22 24
34 33 28 26 43 30 40 32 32 31
25 26 27 34 33 27 33 29 30 31
;
```

A histogram is used to examine the data distribution. For a more complete report, the sample size, minimum value, maximum value, mean, and standard deviation are displayed on the histogram. The following statements illustrate how to inset these statistics:

```
title 'Torsion Strength of Copper Wire';
proc capability data=wire noprint;
   spec lsl=22 llsl=2
        usl=38 lusl=20;
   histogram strength;
   inset n min max mean std;
run;
```

The resulting histogram is displayed in Figure 5.1. The INSET statement immediately follows the plot statement that creates the graphical display (in this case, the HISTOGRAM statement). Specify the keywords for inset statistics (such as N, MIN, MAX, MEAN, and STD) immediately after the word INSET. The inset statistics appear in the order in which you specify the keywords.

A complete list of keywords that you can use with the INSET statement is provided in "Summary of INSET Keywords" on page 198. Note that the set of keywords available for a particular display depends on both the plot statement that precedes the INSET statement and the options that you specify in the plot statement.



Figure 5.1. A Histogram with an Inset

The following examples illustrate options commonly used for enhancing the appearance of an inset.

Formatting Values and Customizing Labels

See CAPINS1 in the SAS/QC Sample Library

By default, each inset statistic is identified with an appropriate label, and each numeric value is printed using an appropriate format. However, you may want to provide your own labels and formats. For example, in Figure 5.1 the default format for the standard deviation prints an excessive number of decimal places. The following statements correct this problem, as well as customizing some of the labels displayed in the inset:

```
proc capability data=wire noprint;
   spec lsl=22 llsl=2
        usl=38 lusl=20;
   histogram strength;
   inset n='Sample Size' min max mean std='Std Dev' (5.2);
run;
```

The resulting histogram is displayed in Figure 5.2. You can provide your own label by specifying the keyword for that statistic followed by an equal sign (=) and the label in quotes. The label can have up to 24 characters.

The format 5.2 specified in parentheses after the keyword STD displays the standard deviation with a field width of five and two decimal places. In general, you can specify any numeric SAS format in parentheses after an inset keyword. You can also specify a format to be used for all the statistics in the INSET statement with the FORMAT= option (see the next example, "Adding a Header and Positioning the Inset"). For more information about SAS formats, refer to *SAS Language Reference: Dictionary*.

Note that if you specify both a label and a format for a statistic, the label must appear before the format, as with the keyword STD in the previous statements.



Figure 5.2. Formatting Values and Customizing Labels in an Inset

Adding a Header and Positioning the Inset

See CAPINS1 in the SAS/QC Sample Library

In the previous examples, the inset is displayed in the upper left corner of the plot, the default position for insets added to histograms. You can control the inset position with the POSITION= option. In addition, you can display a header at the top of the inset with the HEADER= option. The following statements create the chart shown in Figure 5.3:

```
proc capability data=wire noprint;
   spec lsl=22 llsl=2
        usl=38 lusl=20;
   histogram strength;
   inset n='Sample Size' min max range mode sum mean
         std='Std Dev' var stdmean skewness kurtosis
                              /  format   = 6.1
                                 position = rm
                                 header   = 'Data Summary';
run;
```

The header (in this case, *Data Summary*) can be up to 40 characters. Note that a long list of inset statistics is requested. Consequently, POSITION=RM is specified to position the inset in the right margin. For more information about positioning, see "Details" on page 207. Also note that the FORMAT= option is used to format all inset statistics. The *options*, such as HEADER=, POSITION=, and FORMAT=, are specified after the slash (/) in the INSET statement. For more details on INSET statement options, see "Dictionary of Options" on page 204.



Figure 5.3. Adding a Header and Repositioning the Inset

Syntax

The syntax for the INSET statement is as follows:

INSET *keyword-list* < **/** *options* >**;**

You can use any number of INSET statements in the CAPABILITY procedure. Each INSET statement produces an inset and must follow one of the plot statements CDFPLOT, COMPHISTOGRAM, HISTOGRAM, PPPLOT, PROBPLOT, or QQPLOT. The inset appears in all displays produced by the plot statement that immediately precedes it. The statistics are displayed in the order in which they are specified. For example, the following statements produce a cumulative distribution plot with two insets and a histogram with one inset:

```
proc capability data=wire;
   cdfplot strength;
      inset mean std min max n;
      inset p1 p5 p10;
   histogram strength;
      inset var skewness kurtosis;
run;
```

The statistics displayed in an inset are computed for a specific process variable using observations for the current BY group. For example, in the following statements, there are two process variables (STRENGTH and DIAMETER) and a BY variable (BATCH). If there are three different batches (levels of BATCH), then a total of six histograms are produced. The statistics in each inset are computed for a particular variable and batch. The labels in the inset are the same for each histogram.

```
proc capability data=wire2;
   by batch;
   histogram strength diameter / normal;
   inset mean std min max normal(mu sigma);
run;
```

The components of the INSET statement are described as follows.

keyword-list
can include any of the *keywords* listed in "Summary of INSET Keywords" on page 198. Some *keywords* allow *secondary keywords* to be specified in parentheses immediately after the *primary keyword*. Also, some inset statistics are available only if you request plot statements and options for which those statistics are calculated. For example, consider the following statements:

```
proc capability data=wire;
   histogram strength / normal;
   inset mean std normal(ad adpval);
run;
```

The *keywords* MEAN and STD display the sample mean and standard deviation of STRENGTH. The *primary keyword* NORMAL with the *secondary keywords* AD and ADPVAL display the Anderson-Darling goodness-of-fit test statistic and p-value in

the inset as well. The statistics specified with the NORMAL keyword are available only because a normal distribution has been fit to the data using the NORMAL option in the HISTOGRAM statement. See the "Summary of INSET Keywords" section, which follows, for a list of available *keywords*.

Typically, you specify *keywords*, to display statistics computed by the CAPABILITY procedure. However, you can also specify the *keyword* DATA= followed by the name of a SAS data set to display customized statistics. This data set must contain two variables:

- a character variable named _LABEL_ whose values provide labels for inset entries.
- a variable named _VALUE_, which can be either character or numeric, and whose values provide values for inset entries.

The label and value from each observation in the DATA= data set occupy one line in the inset. The position of the DATA= keyword in the keyword list determines the position of its lines in the inset.

By default, inset statistics are identified with appropriate labels, and numeric values are printed using appropriate formats. However, you can provide customized labels and formats. You provide the customized label by specifying the *keyword* for that statistic followed by an equal sign (=) and the label in quotes. Labels can have up to 24 characters. You provide the numeric format in parentheses after the *keyword*. Note that if you specify both a label and a format for a statistic, the label must appear before the format. For an example, see "Formatting Values and Customizing Labels" on page 193.

options

appear after the slash (/) and control the appearance of the inset. For example, the following INSET statement uses two appearance *options* (POSITION= and CTEXT=):

```
inset mean std min max / position=ne ctext=yellow;
```

The POSITION= option determines the location of the inset, and the CTEXT= option specifies the color of the text of the inset.

See "Summary of Options" on page 204 for a list of all available *options*, and "Dictionary of Options" on page 204 for detailed descriptions. Note the difference between *keywords* and *options*; *keywords* specify the information to be displayed in an inset, whereas *options* control the appearance of the inset.

Summary of INSET Keywords

Summary Statistics and Process Capability Indices

Table 5.1. Summary Statistics

N	sample size
SUMWGT	sum of the weights
MEAN	sample mean
SUM	sum of the observations
STD	standard deviation
VAR	variance
SKEWNESS	skewness
KURTOSIS	kurtosis
MAX	largest value
MIN	smallest value
NOBS	number of observations
RANGE	range
MODE	most frequent value
NMISS	number of missing values
USS	uncorrected sum of squares
CSS	corrected sum of squares
CV	coefficient of variation
STDMEAN	standard error of the mean
DATA=	arbitrary values from *SAS-data-set*

Table 5.2. Percentile Statistics

P1	1^{st} percentile
P5	5^{th} percentile
P10	10^{th} percentile
Q1	lower quartile (25^{th} percentile)
MEDIAN	median (50^{th} percentile)
Q3	upper quartile (75^{th} percentile)
P90	90^{th} percentile
P95	95^{th} percentile
P99	99^{th} percentile
QRANGE	interquartile range (Q3 - Q1)

Table 5.3. Test of Normality

NORMALTEST	test statistic for normality
PNORMAL	probability value for the normality test

Table 5.4. Signed Rank Test

SIGNRANK	signed rank statistic
PROBS	probability value for the signed rank test

Table 5.5. Capability Indices and Confidence Limits

CP	capability index C_p
CPLCL	lower confidence limit for C_p
CPUCL	upper confidence limit for C_p
CPK	capability index C_{pk}
CPKLCL	lower confidence limit for C_{pk}
CPKUCL	upper confidence limit for C_{pk}
CPL	capability index *CPL*
CPM	capability index C_{pm}
CPMLCL	lower confidence limit for C_{pm}
CPMUCL	upper confidence interval for C_{pm}
CPU	capability index *CPU*
K	capability index K

Table 5.6. Specification Limits and Related Information

LSL	lower specification limit
USL	upper specification limit
TARGET	target value
PCTGTR	percent of nonmissing observations that exceed the upper specification limit
PCTLSS	percent of nonmissing observations that are less than the lower specification limit
PCTBET	percent of nonmissing observations between the upper and lower specification limits (inclusive)

Table 5.7. Student's t-Test

T	statistic for Student's t-test
PROBT	probability value for Student's t-test

Statistics Available with Parametric Density Estimates

You can request parametric density estimates with all plot statements in the CAPABILITY procedure (CDFPLOT, COMPHISTOGRAM, HISTOGRAM, PPPLOT, PROBPLOT, and QQPLOT). You can display parameters and statistics associated with these estimates in an inset by specifying a distribution keyword followed by secondary keywords in parentheses. For example, the following statements create a histogram for STRENGTH with a fitted exponential density curve:

```
proc capability data=wire;
   histogram strength / exp;
   inset exp(sigma theta);
run;
```

The secondary keywords SIGMA and THETA for the EXP distribution keyword request an inset displaying the values of the exponential scale parameter σ and threshold

parameter θ. You must request the distribution option in the plot statement to display the corresponding distribution statistics in an inset. Specifying a distribution keyword with no secondary keywords produces an inset displaying the full set of parameters for that distribution. See Output 5.1.1 on page 211 for an example of an inset with statistics from a fitted normal curve.

The following table describes the available distribution keywords. Note that some keywords are not available with all plot statements.

Table 5.8. Density Estimation Primary Keywords

Keyword	Distribution	Plot Statement Availability
BETA	beta	all except COMPHISTOGRAM
EXPONENTIAL	exponential	all except COMPHISTOGRAM
GAMMA	gamma	all except COMPHISTOGRAM
LOGNORMAL	lognormal	all except COMPHISTOGRAM
NORMAL	normal	all plot statements
SB	Johnson S_B	all except COMPHISTOGRAM
SU	Johnson S_U	all except COMPHISTOGRAM
WEIBULL	Weibull	all except COMPHISTOGRAM
WEIBULL2	2-parameter Weibull	PROBPLOT and QQPLOT

Table 5.9 through Table 5.17 list the secondary keywords available with each distribution keyword listed in Table 5.8. In many cases, aliases can be used (for example, ALPHA in place of SHAPE1).

Table 5.9. Secondary Keywords Available with the BETA Keyword

Secondary Keyword	Alias	Description
ALPHA	SHAPE1	first shape parameter α
BETA	SHAPE2	second shape parameter β
SIGMA	SCALE	scale parameter σ
THETA	THRESHOLD	lower threshold parameter θ

Table 5.10. Secondary Keywords Available with the EXP Keyword

Secondary Keyword	Alias	Description
SIGMA	SCALE	scale parameter σ
THETA	THRESHOLD	threshold parameter θ

Table 5.11. Secondary Keywords Available with the GAMMA Keyword

Secondary Keyword	Alias	Description
ALPHA	SHAPE	shape parameter α
SIGMA	SCALE	scale parameter σ
THETA	THRESHOLD	threshold parameter θ

Table 5.12. Secondary Keywords Available with the LOGNORMAL Keyword

Secondary Keyword	Alias	Description
SIGMA	SHAPE	shape parameter σ
THETA	THRESHOLD	threshold parameter θ
ZETA	SCALE	scale parameter ζ

Table 5.13. Secondary Keywords Available with the NORMAL Keyword

Secondary Keyword	Alias	Description
MU	MEAN	mean parameter μ
SIGMA	STD	scale parameter σ

Table 5.14. Secondary Keywords Available with the SB Keyword

Secondary Keyword	Alias	Description
DELTA		shape parameter δ
GAMMA		shape parameter γ
SIGMA	SHAPE	scale parameter σ
THETA	THRESHOLD	threshold parameter θ

Table 5.15. Secondary Keywords Available with the SU Keyword

Secondary Keyword	Alias	Description
DELTA		shape parameter δ
GAMMA		shape parameter γ
SIGMA	SHAPE	scale parameter σ
THETA		location parameter θ

Table 5.16. Secondary Keywords Available with the WEIBULL Keyword

Secondary Keyword	Alias	Description
C	SHAPE	shape parameter c
SIGMA	SCALE	scale parameter σ
THETA	THRESHOLD	threshold parameter θ

Table 5.17. Secondary Keywords Available with the WEIBULL2 Keyword

Secondary Keyword	Alias	Description
C	SHAPE	shape parameter c
SIGMA	SCALE	scale parameter σ
THETA	THRESHOLD	known lower threshold θ_0

The secondary keywords listed in Table 5.18 can be used with any distribution keyword but *only* with the HISTOGRAM and COMPHISTOGRAM plot statements.

Table 5.18. Statistics Computed from Any Parametric Density Estimate

Secondary Keyword	Description
CP	capability index C_p
CPK	capability index C_{pk}
CPL	capability index *CPL*
CPM	capability index C_{pm}
CPU	capability index *CPU*
ESTPCTLSS	estimated percentage less than the lower specification limit
ESTPCTGTR	estimated percentage greater than the upper specification limit
K	capability index K

The secondary keywords listed in Table 5.19 can be used with any distribution keyword but *only* with the HISTOGRAM plot statement (see Example 5.1 on page 211).

Table 5.19. Goodness-of-Fit Statistics for Fitted Curves

Secondary Keyword	Description
CHISQ	chi-square statistic
DF	degrees of freedom for the chi-square test
PCHISQ	probability value for the chi-square test
AD	Anderson-Darling EDF test statistic
ADPVAL	Anderson-Darling EDF test p-value
CVM	Cramér-von Mises EDF test statistic
CVMPVAL	Cramér-von Mises EDF test p-value
KSD	Kolmogorov-Smirnov EDF test statistic
KSDPVAL	Kolmogorov-Smirnov EDF test p-value

Table 5.20 lists primary keywords available only with the HISTOGRAM and COMPHISTOGRAM plot statements. These keywords display fill areas on a histogram. If you fit a parametric density on a histogram and request that the area under the curve be filled, these keywords display the percentage of the distribution area that lies below the lower specification limit, between the specification limits, or above the upper specification limit. If you do not fill the area beneath a parametric density estimate, these keywords display the observed proportion of observations (that is, the area in the bars of the histogram).

You should use these options with the FILL, CFILL=, and PFILL= options in the HISTOGRAM and COMPHISTOGRAM statements and with the CLEFT=, CRIGHT=, PLEFT=, and PRIGHT= options in the SPEC statements. See Output 5.2.1 on page 213 for an example.

Table 5.20. Curve Area Keywords

Keyword	Alias	Description
BETWEENPCT	BETPCT	area between the specification limits
LSLPCT		area below the lower specification limit
USLPCT		area above the upper specification limit

Statistics Available with Nonparametric Kernel Density Estimates

You can request nonparametric kernel density estimates with the HISTOGRAM and COMPHISTOGRAM plot statements. You can display statistics associated with these estimates by specifying a kernel density keyword followed by secondary keywords in parentheses. For example, the following statements create a histogram for STRENGTH with a fitted kernel density estimate:

```
proc capability data=wire;
   histogram strength / kernel;
   inset kernel(c amise);
run;
```

The secondary keywords C and AMISE for the KERNEL keyword display the values of the standardized bandwidth c and the approximate mean integrated square error.

Note that you can specify up to five kernel density estimates on a single histogram. If you specify multiple kernel density estimates, you can request inset statistics for all of the estimates with the KERNEL keyword, or you can display inset statistics for individual curves with KERNELn keywords, as in the following example:

```
proc capability data=wire;
   histogram strength / kernel(c = 1 2 3);
   inset kernel2(c) kernel3(c);
run;
```

Three kernel density estimates are displayed on the histogram, but the inset displays the value of c only for the second and third estimates.

Table 5.21 lists the kernel density keywords. Table 5.22 lists the available secondary keywords.

Table 5.21. Kernel Density Estimate Primary Keywords

Keyword	Description
KERNEL	displays statistics for all kernel estimates
KERNELn	displays statistics for only the n^{th} kernel density estimate $n = 1, 2, 3, 4,$ or 5

Table 5.22. Secondary Keywords Available with the KERNEL Keyword

Secondary Keyword	Description
TYPE	kernel type: normal, quadratic, or triangular
BANDWIDTH	bandwidth λ for the density estimate
BWIDTH	alias for BANDWIDTH
C	standardized bandwidth c for the density estimate: $c = \frac{\lambda}{Q} n^{\frac{1}{5}}$ where $n =$ sample size, $\lambda =$ bandwidth, and $Q =$ interquartile range
AMISE	approximate mean integrated square error (MISE) for the kernel density

Summary of Options

The following table lists the INSET statement options. For complete descriptions, see "Dictionary of Options," which follows this section.

Table 5.23. INSET Options

CFILL=*color* | BLANK	specifies color of inset background
CFILLH=*color*	specifies color of header background
CFRAME=*color*	specifies color of frame
CHEADER=*color*	specifies color of header text
CSHADOW=*color*	specifies color of drop shadow
CTEXT=*color*	specifies color of inset text
DATA	specifies data units for POSITION=(x, y) coordinates
FONT=*font*	specifies font of text
FORMAT=*format*	specifies format of values in inset
HEADER='*quoted string*'	specifies header text
HEIGHT=*value*	specifies height of inset text
NOFRAME	suppresses frame around inset
POSITION=*position*	specifies position of inset
REFPOINT=BR|BL|TR|TL	specifies reference point of inset positioned with POSITION=(x, y) coordinates

Dictionary of Options

The following entries provide detailed descriptions of options for the INSET statement. Terms used in this section are illustrated in Figure 5.4.



Figure 5.4. The Inset

CFILL=*color* **| BLANK**
specifies the color of the background (including the header background if you do not specify the CFILLH= option). See Output 5.1.1 on page 211 for an example.

If you do not specify the CFILL= option, then by default, the background is empty. This means that items that overlap the inset (such as curves, histogram bars, or specification limits) show through the inset. If you specify any value for the CFILL= option, then overlapping items no longer show through the inset. Specify CFILL=BLANK to leave the background uncolored and also to prevent items from showing through the inset.

CFILLH=*color*
specifies the color of the header background. By default, if you do not specify a CFILLH= color, the CFILL= color is used.

CFRAME=*color*
specifies the color of the frame. By default, the frame is the same color as the axis of the plot.

CHEADER=*color*
specifies the color of the header text. By default, if you do not specify a CHEADER= color, the CTEXT= color is used.

CSHADOW=*color*
CS=*color*
specifies the color of the drop shadow. See Output 5.2.1 on page 213 for an example. By default, if you do not specify the CSHADOW= option, a drop shadow is not displayed.

CTEXT=*color*
CT=*color*
specifies the color of the text. By default, the inset text color is the same as the other text on the plot.

DATA
specifies that data coordinates are to be used in positioning the inset with the POSITION= option. The DATA option is available only when you specify POSITION= (x, y), and it must be placed immediately after the coordinates (x, y). For details, see the entry for the POSITION= option or "Positioning the Inset Using Coordinates" on page 208. See Figure 5.7 on page 209 for an example.

FONT=*font*
specifies the font of the text. By default, the font is SIMPLEX if the inset is located in the interior of the plot, and the font is the same as the other text displayed on the plot if the inset is located in the exterior of the plot.

FORMAT=*format*
specifies a format for all the values displayed in an inset. If you specify a format for a particular statistic, then this format overrides the format you specified with the FORMAT= option. See Figure 5.3 on page 195 or Output 5.1.1 on page 211 for an example.

HEADER= *'string'*
specifies the header text. The *string* cannot exceed 40 characters. If you do not specify the HEADER= option, no header line appears in the inset. If all the keywords listed in the INSET statement are secondary keywords corresponding to a fitted curve on a histogram, a default header is displayed that indicates the distribution and identifies the curve. See Figure 5.3 on page 195 for an example of a specified header and Output 5.1.1 on page 211 for an example of the default header for a fitted normal curve.

HEIGHT=*value*
specifies the height of the text.

NOFRAME
suppresses the frame drawn around the text.

POSITION=*position*
POS=*position*
determines the position of the inset. The *position* can be a compass point keyword, a margin keyword, or a pair of coordinates (x, y). You can specify coordinates in axis percent units or axis data units. For more information, see "Details" on page 207. By default, POSITION=NW, which positions the inset in the upper left (northwest) corner of the display.

REFPOINT=BR | BL | TR | TL
RP=BR | BL | TR | TL
specifies the reference point for an inset that is positioned by a pair of coordinates with the POSITION= option. Use the REFPOINT= option with POSITION= coordinates. The REFPOINT= option specifies which corner of the inset frame you want positioned at coordinates (x, y). The keywords BL, BR, TL, and TR represent bottom left, bottom right, top left, and top right, respectively. See Figure 5.8 on page 210 for an example. The default is REFPOINT=BL.

If you specify the position of the inset as a compass point or margin keyword, the REFPOINT= option is ignored. For more information, see "Positioning the Inset Using Coordinates" on page 208.

Details

This section provides details on three different methods of positioning the inset using the POSITION= option. With the POSITION= option, you can specify

- compass points
- keywords for margin positions
- coordinates in data units or percent axis units

Positioning the Inset Using Compass Points

You can specify the eight compass points N, NE, E, SE, S, SW, W, and NW as keywords for the POSITION= option. The following statements create the display in Figure 5.5, which demonstrates all eight compass positions. The default is NW.

See CAPINS2 in the SAS/QC Sample Library

```
proc capability data=wire;
   histogram strength / cfill=gray;
   inset n     / cfill=blank header='Position = NW' pos=nw;
   inset mean  / cfill=blank header='Position = N ' pos=n ;
   inset sum   / cfill=blank header='Position = NE' pos=ne;
   inset max   / cfill=blank header='Position = E ' pos=e ;
   inset min   / cfill=blank header='Position = SE' pos=se;
   inset nobs  / cfill=blank header='Position = S ' pos=s ;
   inset range / cfill=blank header='Position = SW' pos=sw;
   inset mode  / cfill=blank header='Position = W ' pos=w ;
run;
```



Figure 5.5. Insets Positioned Using Compass Points

Positioning the Inset in the Margins

You can also position the inset in one of the four margins surrounding the plot area using the margin keywords LM, RM, TM, or BM, as illustrated in Figure 5.6.



Figure 5.6. Positioning Insets in the Margins

For an example of an inset placed in the right margin, see Figure 5.3 on page 195. Margin positions are recommended if a large number of statistics are listed in the INSET statement. If you attempt to display a lengthy inset in the interior of the plot, it is likely that the inset will collide with the data display.

Positioning the Inset Using Coordinates

You can also specify the position of the inset with coordinates: POSITION= (x, y). The coordinates can be given in axis percent units (the default) or in axis data units.

Data Unit Coordinates

See CAPINS2 in the SAS/QC Sample Library

If you specify the DATA option immediately following the coordinates, the inset is positioned using axis data units. For example, the following statements place the bottom left corner of the inset at 12.5 on the horizontal axis and 10 on the vertical axis:

```
proc capability data=wire;
   histogram strength / cfill=gray;
   inset n / header   = 'Position=(12.5,10)'
             position = (12.5,10) data;
run;
```

The histogram is displayed in Figure 5.7. By default, the specified coordinates determine the position of the bottom left corner of the inset. You can change this reference point with the REFPOINT= option, as in the next example.



Figure 5.7. Inset Positioned Using Data Unit Coordinates

Axis Percent Unit Coordinates

See CAPINS2 in the SAS/QC Sample Library

If you do not use the DATA option, the inset is positioned using axis percent units. The coordinates of the bottom left corner of the display are $(0, 0)$, while the upper right corner is $(100, 100)$. For example, the following statements create a histogram with two insets, both positioned using coordinates in axis percent units:

```
proc capability data=wire;
   histogram strength / cfill=gray;
   inset min / position = (5,25)
               header   = 'Position=(5,25)'
               refpoint = tl;
   inset max / position = (95,95)
               header   = 'Position=(95,95)'
               refpoint = tr;
run;
```

The display is shown in Figure 5.8. Notice that the REFPOINT= option is used to determine which corner of the inset is to be placed at the coordinates specified with the POSITION= option. The first inset has REFPOINT=TL, so the top left corner of the inset is positioned 5% of the way across the horizontal axis and 25% of the way up the vertical axis. The second inset has REFPOINT=TR, so the top right corner of the inset is positioned 95% of the way across the horizontal axis and 95% of the way up the vertical axis. Note also that coordinates in axis percent units must be *between* 0 and 100.



Figure 5.8. Inset Positioned Using Axis Percent Unit Coordinates

Examples

This section provides advanced examples using the INSET statement.

Example 5.1. Inset for Goodness-of-Fit Statistics

This example fits a normal curve to the torsion strength data used in the "Getting Started" section on page 192. The following statements fit a normal curve and request an inset summarizing the fitted curve with the mean, the standard deviation, and the Anderson-Darling goodness-of-fit test:

See CAPINS3 in the SAS/QC Sample Library

```
title 'Torsion Strength of Copper Wire';
proc capability data=wire noprint;
   spec lsl=22 llsl=2
        usl=38 lusl=20 ;
   histogram strength / normal(l=34)
                        nocurvelegend;
   inset normal(mu sigma ad adpval) / cfill  = yellow
                                      format = 7.2;
run;
```

The resulting histogram is displayed in Output 5.1.1. The NOCURVELEGEND option in the HISTOGRAM statement suppresses the default legend for curve parameters.

Output 5.1.1. Inset Table with Normal Curve Information



Example 5.2. Inset for Areas Under a Fitted Curve

See CAPINS4 in the SAS/QC Sample Library

You can use the INSET keywords LSLPCT, USLPCT, and BETWEENPCT to inset legends for areas under histogram bars or fitted curves. The following statements create a histogram with an inset legend for the shaded area under the fitted normal curve to the left of the lower specification limit:

```
title 'Torsion Strength of Copper Wire';
proc capability data=wire noprint;
   spec lsl=22 llsl=2   cleft=red
        usl=38 lusl=20 ;
   histogram strength / cfill  = yellow
                        normal(fill);
   inset  lsl='LSL' lslpct / cshadow=black;
run;
```

The histogram is displayed in Output 5.2.1. The LSLPCT keyword in the INSET statement requests a legend for the area under the curve to the left of the lower specification limit. The CLEFT= option is used to fill the area under the normal curve to the left of the line, and the CFILL= color is used to fill the remaining area. If the FILL *normal-option* were not specified, the CLEFT= and CFILL= colors would be applied to the corresponding areas under the histogram, not the normal curve, and the inset box would reflect the area under the histogram bars.

You can use the USLPCT keyword in the INSET statement to request a legend for the area to the right of an upper specification limit, and you can use the BETWEENPCT keyword to request a legend for the area between the lower and upper limits. By default, the legend requested with each of the keywords LSLPCT, USLPCT, and BETWEENPCT displays a rectangle that matches the color of the corresponding area. You can substitute a customized label for each rectangle by specifying the keyword followed by an equal sign (=) and the label in quotes.

Output 5.2.1. Displaying Areas Under the Normal Curve



The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/QC® User's Guide, Version 8*, Cary, NC: SAS Institute Inc., 1999. 1994 pp.

SAS/QC® User's Guide, Version 8

ISBN 1–58025–493–4

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999